Karlsruher Institut für Technologie

Institut für Programmstrukturen
und Datenorganisation (IPD)

Lehrstuhl Prof. Dr.-Ing. Snelting

Formal Verification of Pattern Matching Analyses

Masterarbeit von

Henning Dieterichs

an der Fakultät für Informatik



```
theorem ℛ_semantic : ∀ can_prove_empty: CorrectCanProveEmpty, ∀ gdt: Gdt, gdt.disjoint_rhss →
    (
        let ⟨ a, i, r ⟩ := ℛ can_prove_empty.val (𝒜 gdt)
        in
            -- Reachable rhss are accessible and neither inaccessible nor redundant.
            (∀ env: Env, ∀ rhs: Rhs,
                gdt.eval env = Result.value rhs
                  → rhs ∈ a \ (i ++ r)
            )
          ∧
            -- Redundant rhss can be removed without changing semantics.
            Gdt.eval_option (gdt.remove_rhss r.to_finset)
            = gdt.eval

        : Prop
    )
```

Erstgutachter:	Prof. Dr.-Ing. Gregor Snelting
Zweitgutachter:	Prof. Dr. rer. nat. Bernhard Beckert
Betreuende Mitarbeiter:	M. Sc. Sebastian Graf
Abgabedatum:	**?today?**

Zusammenfassung

Deutsche Übersetzung einfügen

The algorithms presented in Lower Your Guards [1] analyze pattern matching expressions and detect uncovered cases and inaccessible or redundant right hand sides. They already have been implemented in the Glasgow Haskell Compiler and spotted previously unknown bugs in real world code. While these algorithms have been validated empirically, their correctness has not been precisely defined nor proven yet.

This thesis establishes a precise notion of correctness and presents formal proofs that these algorithms are indeed correct. These proofs are formalized in Lean 3.

Contents

1 Introduction

In functional programming, pattern matching is a very popular feature. This is particularly true for Haskell, where you can define algebraic data types and easily match on them in function definitions. With increasingly complex data types and function definitions however, pattern matching can be yet another source of mistakes.

Figure 1.1 showcases common types of mistakes that can arise with pattern matching.

Most importantly, the function `f` is not defined on all values: Evaluating `f Case4` will cause a runtime error! In other words, the pattern match used to define `f` is not exhaustive, as the input `Case4` is uncovered. This is usually an oversight by the programmer and should be brought to their attention with an appropriate warning.

Also, `f` will never evaluate to 3 or 4 - replacing these values with any other value would not change any observable behaviour of `f`. Such right hand sides (*RHS*s) are called *inaccessible*. Inaccessible RHSs indicate a code smell and should be avoided too. Sometimes, such RHSs can simply be removed from the pattern.

Figure 1.1: A Pattern Matching Example In Haskell

```
data Case = Case1 | Case2 | Case3 | Case4

f :: Case -> Bool -> Integer
f Case1 _ = 1
f Case2 _ = 2
f x True | Case1 <- x = 3
         | Case2 <- x = 4
f Case3 _ = 5
```

Lower Your Guards (LYG) [1] is a compiler analysis that is able to detect such mistakes and also can deal with the intricacies of lazy evaluation.

However, LYG is only checked empirically so far: Its implementation in the Glasgow Haskell Compiler just *seems to work*.

Obviously, LYG would be incorrect if it marks a RHS as inaccessible even though it actually is accessible. This could have fatal consequences: A programmer acting on such misinformation might delete a RHS that is very much in use!

As LYG does not give a complete characterization of correctness, we first want to establish a precise notion of correctness and then check that these algorithms indeed comply with it. At the very least, a verifying tool should be verified itself!

The large number of case distinctions made in the algorithms motivates the use of a theorem prover; a natural proof would not be very trustworthy due to the high technical demand and risk of missing edge cases.

The main contributions of this thesis are as follows:

- We formalized the uncovered and redundant/inaccessible analysis of LYG in Lean 3. This formalization is discussed in detail in chapter 3. We noticed an inaccuracy in how variable scopes are handled in refinement types and present a counter-example to $\mathcal{U}$'s correctness in chapter 3.1.3 by exploiting shadowing variable bindings. We suggest a more explicit variable scoping mechanism of refinement types.

- We establish a notion of correctness of LYG. Its formalization in Lean is discussed in chapter 3.2. This notion of correctness is more precise and more complete than the notion of correctness presented in LYG.

- We present formal proofs that the redundant/inaccessible analysis of LYG satisfies this notion of correctness if our suggestion of a more explicit scoping operator is applied. Details of this proof are discussed in chapter 4.

2 Background

2.1 Lower Your Guards

Lower Your Guards (LYG) [1] describes algorithms that analyze pattern matching expressions and report uncovered cases, but also redundant and inaccessible right hand sides.

LYG was designed for use in the Glasgow Haskell Compiler, but the algorithm and its data structures are so universal that they can be leveraged for other programming languages with pattern matching constructs too.

All definitions and some examples of this chapter are taken from LYG [1].

2.1.1 Inaccessible vs. Redundant RHSs

A closer look at figure 1.1 reveals that while both RHS 3 and 4 are inaccessible, the semantics of `f` changes if both are removed. This means that an automated refactoring cannot just remove all inaccessible leaves!

The reason for this is the term $t :=$ `f Case3 undefined` and the fact that Haskell uses a lazy evaluation strategy. If both RHSs 3 and 4 are removed, t evaluates to 5 - the term `undefined` is never evaluated as no pattern matches against it. However, if nothing or only one of the RHSs 3 or 4 is removed, `undefined` will be matched with `True` and thus t will throw a runtime error!

To communicate this difference, LYG introduces the concept of *redundant* and *inaccessible* RHSs: A redundant RHS can be removed from its pattern matching expression without any observable difference. An inaccessible RHS is never evaluated, but its removal might lead to observable changes. This definition implies that redundant RHSs are inaccessible.

As of listing 1.1, LYG will mark RHS 3 as inaccessible and RHS 4 as redundant. This choice is somewhat arbitrary, as RHS 3 could be marked as redundant and RHS 4 as inaccessible as well, and will be discussed in more detail chapter ref.

2.1.2 Guard Trees

For all analyses, LYG first transforms Haskell specific pattern match expressions to simpler *guard trees*. This transformation removes a lot of complexity, as many different Haskell constructs can be desugared to the same guard tree. Guard trees also simplify adapting LYG to other programming languages and they enable studying LYG mostly independent from Haskell. Their syntax is defined in figure 2.1.

Guard trees (*Gdt*s) are made of three elements: Uniquely numbered right hand sides, *branches* and *guarded trees*. Guarded trees refer to Haskell specific guards (*Grd*) that control the execution. *Let guards* can bind a term to a variable in a new lexical scope. *pattern match guards* can destructure a value into variables if the pattern matches or otherwise prevent the execution from entering the tree behind the guard. Finally, *bang guards* can stop the entire execution when the value of a variable does not reduce to a head normal form.

Figure 2.1: Definition of Guard Trees

Guard Syntax

$$
\begin{array}{rlll}
k, n, m \in & \mathbb{N} & & \\
K \in & \mathsf{Con} & & \\
x, y, a, b \in & \mathsf{Var} & & \\
\tau, \sigma \in & \mathsf{Type} & a \mid ... & \\
e \in & \mathsf{Expr} & x \mid K\ \overline{\tau}\ \overline{\gamma}\ \overline{e} \mid ... &
\end{array}
\qquad
\begin{array}{rlll}
\gamma \in & \mathsf{TyCt} & \tau_1 \sim \tau_2 \mid ... & \\
p \in & \mathsf{Pat} & \_ \mid K\ \overline{p} \mid ... & \\
g \in & \mathsf{Grd} & \mathsf{let}\ x : \tau = e & \\
& & \mid\ K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x & \\
& & \mid\ !x &
\end{array}
$$

Guard Tree Syntax

$$t \in \mathsf{Gdt} \qquad \longrightarrow k \mid \top \begin{array}{l} t_1 \\ t_2 \end{array} \mid \dashv g \multimap t$$

The evaluation of a guard tree selects the first right hand side that execution reaches. If the execution stops at a bang guard, the evaluation is said to *diverge*, otherwise, if execution falls through, the evaluation ends with a *no-match*. A formal semantic for guard trees will be defined in chapter 3.1.2.

The transformation from Haskell pattern matches to guard trees is not of much interest for this thesis and can be found in LYG [1]. To preserve semantics, it is important that the transformation inserts bang guards whenever a variable is matched against a data constructor.

Figure 2.2 presents the transformation of figure 1.1 into a guard tree.

It is usually straightforward to define a transformation from pattern matching expressions to guard trees that also preserves uncovered cases and inaccessible and redundant RHSs. This makes guard trees an ideal abstraction for the following analysis steps.

2.1.3 Refinement Types

Refinement types describe vectors of values $x_1, ..., x_n$ that satisfy a given predicate Φ. Their syntax is defined in figure 2.3.

Figure 2.2: Desugaring Example

```
data Case = Case1 | Case2 | Case3 | Case4

f :: Case -> Bool -> Integer
f Case1 _ = 1
f Case2 _ = 2
f x True | Case1 <- x = 3
         | Case2 <- x = 4
f Case3 _ = 5
```



Figure 2.3: Definition of Refinement Types

Γ	$\emptyset \mid \Gamma, x : \tau \mid \Gamma, a$	Context
φ	$\checkmark \mid \times \mid K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x \mid x \not\approx K$	
	$\mid x \approx \bot \mid x \not\approx \bot \mid \mathsf{let}\ x = e$	Literals
Φ	$\varphi \mid \Phi \wedge \Phi \mid \Phi \vee \Phi$	Formula
Θ	$\langle\, \Gamma \mid \Phi \,\rangle$	Refinement type

Operations on Θ

$$
\begin{aligned}
\Phi \mathbin{\dot\wedge} \varphi &= \begin{cases} \varphi_1 \wedge (\varphi_2 \mathbin{\dot\wedge} \varphi) \text{ if } \Phi = \varphi_1 \wedge \varphi_2 \\ \Phi \wedge \varphi \text{ otherwise} \end{cases} \\
\langle\, \Gamma \mid \Phi \,\rangle \mathbin{\dot\wedge} \varphi &= \langle\, \Gamma \mid \varphi \mathbin{\dot\wedge} \varphi \,\rangle \\
\langle\, \Gamma \mid \Phi_1 \,\rangle \cup \langle\, \Gamma \mid \Phi_2 \,\rangle &= \langle\, \Gamma \mid \Phi_1 \vee \Phi_2 \,\rangle
\end{aligned}
$$

Refinement predicates are built from literals ϕ and closed under conjunction and disjunction. The literal $\checkmark$ refers to "true", while $\times$ refers to "false". For example:

$$\begin{array}{rll} \langle\, x{:}\mathit{Bool} \mid \checkmark \,\rangle & \text{denotes} & \{\bot, \mathit{True}, \mathit{False}\} \\ \langle\, x{:}\mathit{Bool} \mid x \not\approx \bot \,\rangle & \text{denotes} & \{\mathit{True}, \mathit{False}\} \\ \langle\, x{:}\mathit{Bool} \mid x \not\approx \bot \wedge \mathit{True} \leftarrow x \,\rangle & \text{denotes} & \{\mathit{True}\} \\ \langle\, mx{:}\mathit{Maybe\ Bool} \mid mx \not\approx \bot \wedge \mathit{Just}\, x \leftarrow mx \,\rangle & \text{denotes} & \mathit{Just}\,\{\bot, \mathit{True}, \mathit{False}, \} \end{array}$$

2.1.4 Binding Mechanism Of Refinement Types

Refinement type literals, such as the let-literal or the pattern-match-literal can bind one or more variables. Unconventionally however, a binding is in scope of a literal if and only if the binding literal is the left operand of a parent conjunction.

Thus, $(\mathsf{let}\ x = y \wedge x \not\approx \bot) \wedge x \not\approx \bot$ is semantically equivalent to $y \not\approx \bot \wedge x \not\approx \bot$. Clearly, $\wedge$ is not associative!

To utilize this behaviour, the operator $\dot{\wedge}$ replaces the rightmost operand of the top conjunction tree of the left argument (figure 2.3).

2.1.5 $\mathcal{G}$enerating Inhabitants

LYG also describes a partial function $\mathcal{G}$ with $\mathcal{G}(\Theta) = \emptyset \Rightarrow (\Theta \text{ denotes } \emptyset)$ for all refinement types Θ. $\mathcal{G}$ is used to $\mathcal{G}$enerate inhabitants of a refinement type to build elaborate error messages and to get a guarantee that a refinement type is empty. A total correct function $\mathcal{G}$ is uncomputable, since refinement types can make use of recursively defined functions! This thesis just assumes that "interesting" computable and correct functions $\mathcal{G}$ exist, so the details of $\mathcal{G}$ as proposed by LYG do not matter. In general, all proposed correctness statements should allow for an empty function $\mathcal{G}$.

2.1.6 Uncovered Analysis

The goal of the uncovered analysis is to detect all cases that are not handled by a given guard tree. Refinement types are used to capture the result of this analysis.

The function $\mathcal{U}(\langle\, \Gamma \mid \checkmark \,\rangle, \cdot)$ in figure 2.4 computes a refinement type that captures all uncovered values for a given guard tree. This refinement type is empty if and only if there are not any uncovered cases. If $\mathcal{G}$ is used to test for emptiness, this already yields an algorithm to test for uncovered cases. It can be verified that the uncovered refinement type of the guard tree in figure 2.2 "semantically" equals $\langle\, x_1{:}\mathit{Case}, x_2{:}\mathit{Bool} \mid x_1 \not\approx \bot \,\dot{\wedge}\, x_1 \not\approx \texttt{Case1} \,\dot{\wedge}\, x_1 \not\approx \texttt{Case2} \,\dot{\wedge}\, x_1 \not\approx \texttt{Case3} \,\rangle$ and denotes $x_1 = \texttt{Case4}$.

2.1.7 Annotated Guard Trees

Annotated guard trees represent simplified guard trees that have been annotated with refinement types Θ. They are made of RHSs, branches and bang nodes. Their

Figure 2.4: Definition of $\mathcal{U}$

$$\boxed{\mathcal{U}(\Theta, t) = \Theta}$$

$$\begin{array}{lcl}
\mathcal{U}(\langle\, \Gamma \mid \Phi \,\rangle,\ \longrightarrow n\,) & = & \langle\, \Gamma \mid \times \,\rangle \\
\mathcal{U}(\Theta,\ \top\!\!\!\!\begin{array}{l} t_1 \\ \bot\ t_2 \end{array}) & = & \mathcal{U}(\mathcal{U}(\Theta, t_1), t_2) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!| !x \mathrel{---} t\,) & = & \mathcal{U}(\Theta \mathbin{\dot\wedge} (x \not\approx \bot), t) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!| \mathsf{let}\ x = e \mathrel{---} t\,) & = & \mathcal{U}(\Theta \mathbin{\dot\wedge} (\mathsf{let}\ x = e), t) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!| K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x \mathrel{---} t\,) & = & \Theta \mathbin{\dot\wedge} (x \not\approx K) \cup \mathcal{U}(\Theta \mathbin{\dot\wedge} (K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x), t)
\end{array}$$

syntax is defined in figure 2.5.

Figure 2.5: Definition of Annotated Guard Trees

$$u \in \mathsf{Ant} \quad \longrightarrow \Theta\, k \mid \top\!\!\!\!\begin{array}{l} u_1 \\ \bot\ u_2 \end{array} \mid \mathrel{---} \Theta \lightning \mathrel{---} u$$

2.1.8 Redundant/Inaccessible Analysis

The goal of the redundant/inaccessible analysis is to report as much RHSs as possible that are redundant or inaccessible. This is done by annotating a guard tree with refinement types and then checking these refinement types for emptiness. If a RHS is associated with an empty refinement type, the RHS is inaccessible and in some circumstances even redundant. The refinement type of a bang node describes all values under which an evaluation will diverge. Figure 2.6 defines a function $\mathcal{A}$ that computes such an annotation for a given guard tree. Figure 2.7 shows the annotated tree of the introductory example in figure 1.1 with abbreviated refinement types.

Such an annotated guard tree is then passed to a function $\mathcal{R}$ as defined in figure 2.8. $\mathcal{R}$ uses $\mathcal{G}$ to compute redundant and inaccessible RHSs. All other RHSs are assumed to be accessible, even though, due to $\mathcal{G}$ being a partial function, not all of them actually are accessible.

Figure 2.9 computes inaccessible and redundant leaves for an annotated tree that is ($\mathcal{G} = \emptyset$)-equivalent to the annotated tree from figure 2.7 for sensible functions $\mathcal{G}$. It states that RHS 4 in 1.1 is redundant and can be removed, while RHS 3 is just inaccessible.

Figure 2.6: Definition of $\mathcal{A}$

$$\boxed{\mathcal{A}(\Theta, t) = u}$$

$$\begin{array}{lcl}
\mathcal{A}(\Theta,\ \longrightarrow n\,) & = & \longrightarrow \Theta\, n \\
\mathcal{A}(\Theta,\ \top\!\!\begin{array}{l} t_1 \\ \bot\ t_2 \end{array}) & = & \top\!\!\begin{array}{l} \mathcal{A}(\Theta, t_1) \\ \bot\ \mathcal{A}(\mathcal{U}(\Theta, t_1), t_2) \end{array} \\
\mathcal{A}(\Theta,\ \dashv !x \mathrel{—} t\,) & = & \mathrel{—}\Theta \mathbin{\dot\wedge} (x \approx \bot) \lightning \mathrel{—} \mathcal{A}(\Theta \mathbin{\dot\wedge} (x \not\approx \bot), t) \\
\mathcal{A}(\Theta,\ \dashv \mathsf{let}\ x = e \mathrel{—} t\,) & = & \mathcal{A}(\Theta \mathbin{\dot\wedge} (\mathsf{let}\ x = e), t) \\
\mathcal{A}(\Theta,\ \dashv K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x \mathrel{—} t\,) & = & \mathcal{A}(\Theta \mathbin{\dot\wedge} (K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x), t)
\end{array}$$

Figure 2.7: Examplary Evaluation of $\mathcal{A}$



Figure 2.8: Definition of $\mathcal{R}$. $\mathcal{R}$ partitions all RHSs into could-be-accessible ($\overline{k}$), inaccessible ($\overline{n}$) and $\mathcal{R}$edundant ($\overline{m}$) RHSs.

$$\boxed{\mathcal{R}(u) = (\overline{k}, \overline{n}, \overline{m})}$$

$$\begin{array}{lcl}
\mathcal{R}(\ \longrightarrow \Theta\, n\,) & = & \begin{cases} (\epsilon, \epsilon, n), & \text{if } \mathcal{G}(\Theta) = \emptyset \\ (n, \epsilon, \epsilon), & \text{otherwise} \end{cases} \\
\mathcal{R}(\ \top\!\!\begin{array}{l} t \\ \bot\ u \end{array}) & = & (\overline{k}\,\overline{k'}, \overline{n}\,\overline{n'}, \overline{m}\,\overline{m'}) \text{ where } \begin{array}{l} (\overline{k}, \overline{n}, \overline{m}) = \mathcal{R}(t) \\ (\overline{k'}, \overline{n'}, \overline{m'}) = \mathcal{R}(u) \end{array} \\
\mathcal{R}(\ \mathrel{—}\Theta \lightning \mathrel{—} t\,) & = & \begin{cases} (\epsilon, m, \overline{m'}), & \text{if } \mathcal{G}(\Theta) \neq \emptyset \text{ and } \mathcal{R}(t) = (\epsilon, \epsilon, m\,\overline{m'}) \\ \mathcal{R}(t), & \text{otherwise} \end{cases}
\end{array}$$

Figure 2.9: Examplary Evaluation of $\mathcal{R}$



2.2 Lean

2.2.1 The Lean Theorem Prover

Lean is an interactive theorem prover that is based on the calculus of inductive constructions [2] [3] and is developed by Microsoft Research. It features dependent types, offers a high degree of automation through tactics and can also be used as a programming language. Due to the Curry-Howard isomorphism, writing functional definitions intended to be used in proofs, writing proofs and writing proof-generating custom tactics is very similar.

We use Lean 3 for this thesis and want to give a brief overview of its syntax. See [4] for a detailed documentation.

Inductive data types can be defined with the keyword `inductive`. The `#check` instruction can be used to type-check terms:

```
inductive my_nat : Type
| zero : my_nat
| succ : my_nat → my_nat

#check my_nat.succ my_nat.zero
#check my_nat.zero.succ -- equivalent term, using dot notation
```

The keyword `def` can be used to bind terms and define recursive functions:

```
def my_nat.add : my_nat → my_nat → my_nat
-- Patterns can be used in definitions
| my_nat.zero b := b
| (my_nat.succ a) b := (a.add b).succ
```

Likewise, `def` can be used to bind proof terms to propositions. Propositions are stated as type and proved by constructing a term of that type. Π-types are used to introduce generalized type variables:

```
-- This type states that for all a, a + zero = a
def my_nat.add_zero_eq : Π a: my_nat, a.add my_nat.zero = a :=
    -- Proof by induction
    @my_nat.rec
        -- Induction Hypothesis
        (λ a, a.add my_nat.zero = a)
        -- Case Zero
        (my_nat.add.equations._eqn_1 my_nat.zero)
        -- Case Succ
        (λ a h,
            @eq.subst my_nat
                (λ x, (my_nat.succ a).add my_nat.zero = x.succ)
                (a.add my_nat.zero)
```

```
            a
            h
            (my_nat.add.equations._eqn_2 a my_nat.zero)
    )
```

Proofs are usually much shorter when using Leans tactic mode. Also, definitions can be parametrized (which generalizes the parameter) and the keywords `lemma` and `theorem` can be used instead of `def`:

```
lemma my_nat.add_zero' (a: my_nat): a.add my_nat.zero = a :=
begin
    induction a,
    { refl, },
    { simp [my_nat.add, *], },
end
```

2.2.2 The Lean Mathematical Library

Mathlib [5] is a community project that offers a rich mathematical foundation for many theories in Lean 3. Its theories of finite sets, lists, boolean logic and permutations have been very useful for this thesis.

Mathlib also offers many advanced tactics like `finish`, `tauto` or `linarith`. These tactics help significantly in proving trivial lemmas.

3 Formalization

Before any property of LYG can be proven or even stated in Lean, all relevant definitions must be formalized. Since nothing can be left vague in Lean, a lof of decisions had to be made to back up LYG by a fully defined model. This chapter discusses these decisions.

3.1 Definitions

3.1.1 Abstracting LYG: The Guard Module

LYG does not specify an exact guard or expression syntax. Instead, the notation "..." is often used to indicate a sensible continuation to make guards powerful enough to model all Haskell constructs. This is rather problematic for a precise formalization and presented the first big challenge of this thesis. As we wanted to avoid formalizing Haskell and its semantics, we had to carefully design an abstraction that is as close as possible to LYG while pinning down guards to a closed but extendable theory.

The Result Monad

First, we defined a generic `Result` monad to capture the result of an evaluation. Due to laziness, evaluation of guard trees can either end with a specific right hand side, not match any guard or diverge:

```
inductive Result (α: Type)
| value: α → Result
| diverged: Result
| no_match: Result
```

A `bind` operation can be easily defined on `Result` to make it a proper monad with `Result.value` as unit function:

```
def Result.bind { α β: Type } (f: α → Result β): Result α → Result β
| (Result.value val) := f val
| Result.diverged := Result.diverged
| Result.no_match := Result.no_match
```

Denotational Semantic for Guards

For some abstract environment type `Env`, we would like to have a denotational semantic `Grd.eval` for guards `Grd`:

```
Grd.eval : Grd → Env → Result Env
```

Abstracting `Grd.eval` would unify all guard constructs available in Haskell and those used by LYG. However, LYG needs to recognize all guards that can lead to a diverged evaluation: Removing all RHSs behind such a guard would inevitably remove the guard itself. As this might change the semantic of the guard tree, LYG cannot mark all such RHSs as redundant unless there is a proof that the guard will never diverge. As a consequence, `Grd.eval` cannot be abstracted away.

Instead, we explicitly distinguished between non-diverging (*total*) `tgrd`s and non-no-matching `bang` guards:

```
inductive Grd
| tgrd (tgrd: TGrd)
| bang (var: Var)
```

While `TGrd`s classically represent guards and `Grd`s represent guards with side effects in this context, we decided to follow the naming conventions of LYG and chose the name `TGrd` for side-effect free (non-diverging) guards rather than renaming `Grd`.

In order to define a denotational semantic on `Grd`, we postulated the functions `tgrd_eval : TGrd → Env → option Env` and `is_bottom : Var → Env → bool` as well as a type `Var` that represents variables. While `tgrd`s can change the environment, `bang` guards cannot:

```
def Grd.eval : Grd → Env → Result Env
| (Grd.tgrd grd) env :=
    match tgrd_eval grd env with
    | none := Result.no_match
    | some env' := Result.value env'
    end
| (Grd.bang var) env :=
    if is_bottom var env
    then Result.diverged
    else Result.value env
```

Alternatively, we can set `Var := Env → bool` and `TGrd := Env → bool` and replace `is_bottom` and `tgrd_eval` with `id`, yielding the following definition:

```
inductive Grd'
| tgrd (grd: Env → option Env)
| bang (test: Env → bool)
```

However, this could make the set of guard trees and refinement types uncountable. While this is not problematic for aspects explored by this thesis, it could make implementing a correct function $\mathcal{G}$ impossible, as it cannot reason anymore about guards in a computable way if `Env` is instantiated with a non-finite type.

The Guard Module

In Lean, type classes provide an ideal mechanism to define such ambient abstractions. They can be opened, so that all members of the type class become implicitly available in all definitions and theorems. Every implicit usage pulls the type class into its signature so that consumers can provide a concrete implementation of the type class.

We defined and opened a type class *GuardModule* that describes the presented abstraction:

```
class GuardModule :=
    (Rhs : Type)
    [rhs_decidable: decidable_eq Rhs]
    [rhs_inhabited: inhabited Rhs]
    (Env : Type)
    (TGrd : Type)
    [tgrd_inhabited: inhabited TGrd]
    (tgrd_eval : TGrd → Env → option Env)
    (Var : Type)
    [var_inhabited: inhabited Var]
    (is_bottom : Var → Env → bool)

variable [GuardModule]
open GuardModule
```

We also postulated a type `Rhs` to refer to right hand sides. For technical reasons, equality on this type must be decidable. This abstracts from the numbers that are used in LYG to distinguish right hand sides. We also require most types to be inhabited so that we can construct module-independent examples.

All following definitions and theorems implicitly make use of this abstraction.

3.1.2 Guard Trees

Syntax of Guard Trees

With the definition of `Grd`, guard trees are defined as inductive data type:

```
inductive Gdt
| rhs (rhs: Rhs)
| branch (tr1: Gdt) (tr2: Gdt)
| grd (grd: Grd) (tr: Gdt)
```

Semantic of Guard Trees

`Gdt.eval` defines a denotational semantic on guard trees, using the semantic of guards. It returns the first RHS that matches a given environment. If a guard diverges, the entire evaluation diverges. Otherwise, if no RHSs matches, *no-match* is returned.

```
def Gdt.eval : Gdt → Env → Result Rhs
| (Gdt.rhs rhs) env := Result.value rhs
| (Gdt.branch tr1 tr2) env :=
    match tr1.eval env with
    | Result.no_match := tr2.eval env
    | r := r
    end
| (Gdt.grd grd tr) env := (grd.eval env).bind tr.eval
```

RHSs in Guard Trees

Every guard tree contains a (non-empty) finite set of right hand sides:

```
def Gdt.rhss: Gdt → finset Rhs
| (Gdt.rhs rhs) := { rhs }
| (Gdt.branch tr1 tr2) := tr1.rhss ∪ tr2.rhss
| (Gdt.grd grd tr) := tr.rhss
```

In LYG, it is implicitly assumed that the right hand sides of a guard tree are numbered unambiguously. This has to be stated explicitly in Lean with the following recursive predicate:

```
def Gdt.disjoint_rhss: Gdt → Prop
| (Gdt.rhs rhs) := true
| (Gdt.branch tr1 tr2) :=
        disjoint tr1.rhss tr2.rhss
        ∧ tr1.disjoint_rhss ∧ tr2.disjoint_rhss
| (Gdt.grd grd tr) := tr.disjoint_rhss
```

Removing RHSs in Guard Trees

`Gdt.remove_rhss` defines how a set of RHSs can be removed from a guard tree. This definition is required to state that all redundant RHSs can be removed without changing semantics. Note that the resulting guard tree might be empty when all RHSs are removed!

```
def Gdt.branch_option : option Gdt → option Gdt → option Gdt
| (some tr1) (some tr2) := some (Gdt.branch tr1 tr2)
| (some tr1) none := some tr1
```

```
| none (some tr2) := some tr2
| none none := none

def Gdt.grd_option : Grd → option Gdt → option Gdt
| grd (some tr) := some (Gdt.grd grd tr)
| _ none := none

def Gdt.remove_rhss : finset Rhs → Gdt → option Gdt
| rhss (Gdt.rhs rhs) := if rhs ∈ rhss then none else some (Gdt.rhs rhs)
| rhss (Gdt.branch tr1 tr2) :=
    Gdt.branch_option
        (tr1.remove_rhss rhss)
        (tr2.remove_rhss rhss)
| rhss (Gdt.grd grd tr) := Gdt.grd_option grd (tr.remove_rhss rhss)
```

Finally, to deal with the semantic of empty guard trees, `Gdt.eval_option` lifts `Gdt.eval` to `option Gdt`:

```
def Gdt.eval_option : option Gdt → Env → Result
| (some gdt) env := gdt.eval env
| none env := Result.no_match
```

3.1.3 Refinement Types

Refinement types presented another challenge. Defining refinement types through a proper type system would have required to model some Haskell types. Instead, we tried to rely on the same abstractions used to define guard trees in hope that guard trees and refinement types can be related.

In this formalization, a refinement type Φ denotes a predicate on environments:

```
def Φ.eval: Φ → Env → bool
```

With a proper `GuardModule` instantiation, the environment can be used to not only carry runtime values, but also their type! A (well) typed environment can assist in proving a refinement type to be empty.

Variable Binding Rules

Another problem that had to be solved was the formalization of the unconventional binding mechanism of refinement types through conjunctions, as described in chapter 2.1.4. In particular, this causes $\mathcal{U}$ to be not correct (for some intuitive notion of correctness) with regards to the guard tree semantic we defined in chapter 3.1.2. While the following guard tree *gdt* does not match for any environment, its uncovered refinement type Θ computed by $\mathcal{U}$ is empty (the empty vector does not match)!

$$gdt := \quad \dashv \mathsf{let}\ x = \texttt{False} \begin{cases} \dashv \mathsf{let}\ x = \texttt{True},\ \texttt{False} \leftarrow x \longrightarrow 1 \\ \dashv \texttt{False} \leftarrow x \longrightarrow 2 \end{cases}$$

$$\begin{aligned} \Theta := \mathcal{U}(\checkmark, gdt) &= \langle \mid (((\mathsf{let}\ x = \mathit{False} \mathbin{\dot\wedge} \mathsf{let}\ x = \mathit{True}) \mathbin{\dot\wedge} x \not\approx \mathit{False}) \mathbin{\dot\wedge} x \not\approx \mathit{False} \rangle \\ &= \langle \mid \mathsf{let}\ x = \mathit{False} \wedge (\mathsf{let}\ x = \mathit{True} \wedge (x \not\approx \mathit{False} \wedge x \not\approx \mathit{False})) \rangle \end{aligned}$$

Shadowing is unproblematic for the semantic of guard trees though: If the first guard tree of a branch fails to match, its environment just before the failing guard is discarded and with it possible shadowing bindings. The second branch is always evaluated with the same environment that the first guard tree has been evaluated with.

This imbalance between refinement types and the semantic of guard trees could be fixed by either adjusting the semantics of guard trees or by introducing a guard operator for refinement types with a restricted binding scope. However, if the semantic of guard trees would not undo the effect of no-matching branches to the environment, an inner let binding would prevent an inaccessible right hand side from being redundant. Even though it never matches, the let binding would always have an effect on the final environment and removing it would be observable!

This problem does not arise in the GHC implementation of LYG as it uses a different encoding for refinement types.

We introduced a data constructor `Φ.tgrd_in : TGrd → Φ → Φ` that limits the scope of the guard to the nested refinement type and removed any scoping behaviour of the $\wedge$-operator. This simplifies the scoping mechanism and allows to fix the problem of shadowing bindings in $\mathcal{U}$.

Syntax of Refinement Types

Finally, this is our formalized syntax of refinement types:

```
inductive Φ
| false
| true
| tgrd_in (tgrd: TGrd) (ty: Φ)
| not_tgrd (tgrd: TGrd)
| var_is_bottom (var: Var)
| var_is_not_bottom (var: Var)
| or (ty1: Φ) (ty2: Φ)
| and (ty1: Φ) (ty2: Φ)
```

Since the negation of a guard cannot bind variables, it does not need to have a nested refinement type that would see bound variables. The same applies to `var_is_bottom` and its negation.

Semantic of Refinement Types

The semantic of refinement types is easily defined and implicitly uses the guard module:

```
def Φ.eval: Φ → Env → bool
| Φ.false env := ff
| Φ.true env := tt
| (Φ.tgrd_in grd ty) env := match tgrd_eval grd env with
    | some env := ty.eval env
    | none := ff
    end
| (Φ.not_tgrd grd) env :=
    match tgrd_eval grd env with
    | some env := ff
    | none := tt
    end
| (Φ.var_is_bottom var) env := is_bottom var env
| (Φ.var_is_not_bottom var) env := !is_bottom var env
| (Φ.or t1 t2) env := t1.eval env || t2.eval env
| (Φ.and t1 t2) env := t1.eval env && t2.eval env
```

With this definition the evaluation of the second operand of a conjunction is obviously independent of any environment effects applied in the evaluation of the first operand!

Definition of `is_empty`

A refinement type Φ is called *empty* if it does not match any environment. This is formalized by the predicate Φ.`is_empty`:

```
def Φ.is_empty (ty: Φ): Prop := ∀ env: Env, ¬(ty.eval env)
```

Definition of `can_prove_empty`

Instead of a partial function $\mathcal{G}$ with $\mathcal{G}(\Phi) = \emptyset$ if and only if Φ is empty, we define a total function `can_prove_empty` and a predicate `correct_can_prove_empty` that ensures its correctness. This abstracts from the in this context unneeded generation of inhabitants. It also avoids dealing with partial functions, which are not supported by Lean.

```
variable can_prove_empty: Φ → bool
def correct_can_prove_empty : Prop :=
    ∀ ty: Φ, can_prove_empty ty = tt → ty.is_empty
```

The subtype `CorrectCanProveEmpty` bundles a correct `can_prove_empty` function:

```
def CorrectCanProveEmpty := {
    can_prove_empty : Φ → bool
    // correct_can_prove_empty can_prove_empty
}
```

3.1.4 $\mathcal{U}$ncovered Analysis

As discussed in chapter 3.1.3, LYG's definition of $\mathcal{U}$ has problems with guard trees that define shadowing bindings. LYG defined $\mathcal{U}$ as follows (see chapter 2.1.6 for the discussion of this definition):

$$
\begin{array}{lcl}
\mathcal{U}(\langle\, \Gamma \mid \Phi \,\rangle,\ \longrightarrow n\,) & = & \langle\, \Gamma \mid \times \,\rangle \\
\mathcal{U}(\Theta,\ \top\!\!\!\top^{t_1}_{t_2}) & = & \mathcal{U}(\mathcal{U}(\Theta, t_1), t_2) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!\!| !x \longrightarrow t\,) & = & \mathcal{U}(\Theta \mathbin{\dot\wedge} (x \not\approx \bot), t) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!\!| \mathsf{let}\ x = e \longrightarrow t\,) & = & \mathcal{U}(\Theta \mathbin{\dot\wedge} (\mathsf{let}\ x = e), t) \\
\mathcal{U}(\Theta,\ \longrightarrow\!\!\!| K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x \longrightarrow t\,) & = & \Theta \mathbin{\dot\wedge} (x \not\approx K) \cup \mathcal{U}(\Theta \mathbin{\dot\wedge} (K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x), t)
\end{array}
$$

Equipped with the data constructor `Φ.tgrd_in` we can fix the shadowing problem problem and formalize $\mathcal{U}$ now. Instead of using Φ as accumulator type, our formalization uses the type $\Phi \to \Phi$: The new accumulator explicitly applies a context to a refinement type. This happens implicitly in LYG's definition through the use of $\Theta \mathbin{\dot\wedge} \cdot$. Note that all occuring accumulator functions are homomorphic under the semantic of refinement types. We carefully make use of this to get formalized definitions of $\mathcal{U}$ and $\mathcal{A}$ that can be interleaved, as done in LYG.

```
def 𝒰_acc : (Φ → Φ) → Gdt → Φ
| acc (Gdt.rhs _) := Φ.false
| acc (Gdt.branch tr1 tr2) := (𝒰_acc ((𝒰_acc acc tr1).and ∘ acc) tr2)
| acc (Gdt.grd (Grd.bang var) tr) :=
    𝒰_acc (acc ∘ (Φ.var_is_not_bottom var).and) tr
| acc (Gdt.grd (Grd.tgrd grd) tr) :=
            (acc (Φ.not_tgrd grd))
        .or (𝒰_acc (acc ∘ (Φ.tgrd_in grd)) tr)

def 𝒰 : Gdt → Φ := 𝒰_acc id
```

3.1.5 $\mathcal{R}$edundant / Inaccessible Analysis

Formalization of Annotated Trees

The formalization of annotated trees is straightforward. However, we allow arbitrary annotations rather than only accepting refinement types. This will become useful in formal proofs when we no longer care about the specific refinement types but only if they are empty.

```
inductive Ant (α: Type)
| rhs (a: α) (rhs: Rhs): Ant
| branch (tr1: Ant) (tr2: Ant): Ant
| diverge (a: α) (tr: Ant): Ant
```

Formalization of $\mathcal{A}$

Similar to the formalization of $\mathcal{U}$ in chapter 3.1.4, we also need to address the shadowing problem when formalizing $\mathcal{A}$. This is LYG's definition of $\mathcal{A}$ as stated in chapter 2.1.8:

$$
\begin{array}{lcl}
\mathcal{A}(\Theta,\ \longrightarrow n\ ) & = & \longrightarrow \Theta\, n \\
\mathcal{A}(\Theta,\ \top^{t_1}_{t_2}\ ) & = & \top^{\mathcal{A}(\Theta, t_1)}_{\mathcal{A}(\mathcal{U}(\Theta, t_1), t_2)} \\
\mathcal{A}(\Theta,\ \dashv !x \mathrel{—} t\ ) & = & \mathrel{—} \Theta \mathbin{\dot\wedge} (x \approx \bot) \lightning \mathrel{—} \mathcal{A}(\Theta \mathbin{\dot\wedge} (x \not\approx \bot), t) \\
\mathcal{A}(\Theta,\ \dashv \mathsf{let}\ x = e \mathrel{—} t\ ) & = & \mathcal{A}(\Theta \mathbin{\dot\wedge} (\mathsf{let}\ x = e), t) \\
\mathcal{A}(\Theta,\ \dashv K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x \mathrel{—} t\ ) & = & \mathcal{A}(\Theta \mathbin{\dot\wedge} (K\ \overline{a}\ \overline{\gamma}\ \overline{y : \tau} \leftarrow x), t)
\end{array}
$$

Our formalization in Lean follows. Analogous to our formalization of $\mathcal{U}$, instead of contextualizing refinement types by combining them with the accumulator through $\dot\wedge$, we model the accumulator as an explicit function that contextualizes its argument:

```
def 𝒜_acc : (Φ → Φ) → Gdt → Ant Φ
| acc (Gdt.rhs rhs) := Ant.rhs (acc Φ.true) rhs
| acc (Gdt.branch tr1 tr2) :=
    Ant.branch
        (𝒜_acc acc tr1)
        (𝒜_acc ((𝒰_acc acc tr1).and ∘ acc) tr2)
| acc (Gdt.grd (Grd.bang var) tr) :=
    Ant.diverge
        (acc (Φ.var_is_bottom var))
        (𝒜_acc (acc ∘ ((Φ.var_is_not_bottom var).and)) tr)
| acc (Gdt.grd (Grd.tgrd grd) tr) :=
    (𝒜_acc (acc ∘ (Φ.tgrd_in grd)) tr)

def 𝒜 : Gdt → Ant Φ := 𝒜_acc id
```

Note that in the branch case, `𝒜_acc` and `𝒰_acc` are called with the same arguments. Even more, both functions have the same recursion structure, which makes it possible to interleave both functions. This is done in chapter 3.1.6.

Formalization of $\mathcal{R}$

It remains to formalize the function $\mathcal{R}$ that partitions all right hand sides of an annotated guard tree into accessible, inaccessible and redundant right hand sides, by using the function `can_prove_empty`.

This is $\mathcal{R}$ as presented in LYG and chapter 2.1.8:

$$
\begin{aligned}
\mathcal{R}(\longrightarrow \Theta\, n) &= \begin{cases} (\epsilon, \epsilon, n), & \text{if } \mathcal{G}(\Theta) = \emptyset \\ (n, \epsilon, \epsilon), & \text{otherwise} \end{cases} \\
\mathcal{R}(\top\!\!\!\!\bigsqcup_{u}^{t}) &= (\overline{k}\,\overline{k'}, \overline{n}\,\overline{n'}, \overline{m}\,\overline{m'}) \text{ where } \begin{matrix} (\overline{k}, \overline{n}, \overline{m}) & = \mathcal{R}(t) \\ (\overline{k'}, \overline{n'}, \overline{m'}) & = \mathcal{R}(u) \end{matrix} \\
\mathcal{R}(\text{—}\Theta\lightning\text{—}\, t) &= \begin{cases} (\epsilon, m, \overline{m'}), & \text{if } \mathcal{G}(\Theta) \neq \emptyset \text{ and } \mathcal{R}(t) = (\epsilon, \epsilon, m\,\overline{m'}) \\ \mathcal{R}(t), & \text{otherwise} \end{cases}
\end{aligned}
$$

This definition has a surprisingly direct representation in Lean:

```
def 𝓡 : Ant Φ → list Rhs × list Rhs × list Rhs
| (Ant.rhs ty n) :=
    if can_prove_empty ty
    then ([], [], [n])
    else ([n], [], [])
| (Ant.branch tr1 tr2) :=
    match (𝓡 tr1, 𝓡 tr2) with
    | ((k, n, m), (k', n', m')) := (k ++ k', n ++ n', m ++ m')
    end
| (Ant.diverge ty tr) :=
    match 𝓡 tr, can_prove_empty ty with
    | ([], [], m :: ms), ff := ([], [m], ms)
    | r, _ := r
    end
```

3.1.6 Interleaving $\mathcal{U}$ and $\mathcal{A}$

Since $\mathcal{A}$`_acc` and $\mathcal{U}$`_acc` have the same recursion structure, they can be combined into a single function that shares the recursive invocations. The following function $\mathcal{UA}$`_acc` computes the uncovered refinement type and the annotated guard tree for a given guard tree at the same time. This improves performance if a lazy evaluation strategy is used in combination with sharing as the accumulator can be fully shared.

```
def 𝓤𝓐_acc : (Φ → Φ) → Gdt → Φ × Ant Φ
| acc (Gdt.rhs rhs) := (Φ.false, Ant.rhs (acc Φ.true) rhs)
| acc (Gdt.branch tr1 tr2) :=
    let (U1, A1) := 𝓤𝓐_acc acc tr1,
```

```
        (U2, A2) := 𝒰𝒜_acc (U1.and ∘ acc) tr2
    in  (U2, Ant.branch A1 A2)
| acc (Gdt.grd (Grd.bang var) tr) :=
    let (U, A) := 𝒰𝒜_acc (acc ∘ (Φ.var_is_not_bottom var).and) tr
    in (U, Ant.diverge (acc (Φ.var_is_bottom var)) A)
| acc (Gdt.grd (Grd.tgrd grd) tr) :=
    let (U, A) := 𝒰𝒜_acc (acc ∘ (Φ.tgrd_in grd)) tr
    in ((acc (Φ.not_tgrd grd)).or U, A)
```

It is surprisingly easy to show that this function is really interleaving `𝒜_acc` and `𝒰_acc`:

```
theorem 𝒰𝒜_acc_eq (acc: Φ → Φ) (gdt: Gdt):
    𝒰𝒜_acc acc gdt = (𝒰_acc acc gdt, 𝒜_acc acc gdt) :=
by induction gdt generalizing acc;
    try { cases gdt_grd }; simp [𝒰𝒜_acc, 𝒰_acc, 𝒜_acc, *]
```

3.2 Correctness Statements

As we have all the required definitions at this point, we can state and formalize what we expect of the presented pattern match analyses to be considered correct. We provide proofs for all correctness propositions on GitHub [6]. Chapter 4 will discuss parts of these proofs in more detail.

3.2.1 Correctness of the $\mathcal{U}$ncovered Analysis

$\mathcal{U}$ should compute a refinement type that denotes exactly all values that are not covered by a given guard tree. This does not include values under which the execution diverges!

```
theorem 𝒰_semantic: ∀ gdt: Gdt, ∀ env: Env,
        (𝒰 gdt).eval env ↔ (gdt.eval env = Result.no_match)
```

As an obvious consequence, a guard tree always matches (or diverges) if and only if this refinement type is empty. If a correct function $\mathcal{G}$ or `can_prove_empty` proves emptiness of such a computed refinement type, there clearly are no uncovered cases. Otherwise, a warning of potential uncovered cases should be issued! Hence, this theorem implies correctness of the uncovered analysis: The uncovered analysis should rather report a false positive than not detect an uncovered case.

Note that this theorem carries over to all semantically equivalent definitions of $\mathcal{U}$.

3.2.2 Correctness of the $\mathcal{R}$edundant/Inaccessible Analysis

For a given guard tree and a given correct function `can_prove_empty` (which corresponds to $\mathcal{G}$ in LYG), $\mathcal{R}$ should compute a triple (a, i, r) of accessible, inaccessible and redundant right hand sides. Whenever the given guard tree evaluates to a RHS, this RHS must be accessible and neither inaccessible nor redundant. RHSs that are redundant can be removed without changing the semantic of the guard tree. This expresses correctness of the redundant and inaccessible analysis.

```
theorem ℛ_semantic:
    ∀ can_prove_empty: CorrectCanProveEmpty,
    ∀ gdt: Gdt, gdt.disjoint_rhss → (
        let ⟨ a, i, r ⟩ := ℛ can_prove_empty.val (𝒜 gdt)
        in
                (∀ env: Env, ∀ rhs: Rhs,
                    gdt.eval env = Result.value rhs
                      → rhs ∈ a \ (i ++ r)
                )
            ∧
                Gdt.eval_option (gdt.remove_rhss r.to_finset)
                = gdt.eval

        : Prop
    )
```

Note that redundant RHSs could be marked as inaccessible or even accessible instead without violating this theorem. The opposite is not true: Not all accessible RHSs can be marked as inaccessible and not all inaccessible RHSs can be marked as redundant - see chapters 1 and 2 for counterexamples. However, we conjecture that a contains no inaccessible and i no redundant RHSs if `can_prove_empty` is both correct and complete.

4 Formalized Proofs

This chapter gives an overview of the formal proofs of the correctness statements from the previous chapter. The full Lean proofs can be found on GitHub [6].

To reduce the complexity of the definitions from chapter 3, we came up with several internal definitions. They include accumulator-free alternatives `U` and `A` for the functions $\mathcal{U}$ and $\mathcal{A}$.

Correctness of `U` can be shown directly and this result can be transferred easily to $\mathcal{U}$ too, as $\mathcal{U}$'s correctness only depends on the semantic of the computed refinement type (see chapter 3.2.1). It is much more difficult to show correctness of $\mathcal{R}/\mathcal{A}$ though, so we will discuss this in more detail.

In chapter 4.2, we show that redundant RHSs can be removed without changing semantics. Then, in chapter 4.3, we show that if a guard tree evaluates to a RHS, this RHS must be marked as accessible. Together, these properties form the correctness statement as presented in chapter 3.2.2.

4.1 Simplification A of $\mathcal{A}$

It is difficult to reason about $\mathcal{A}\_acc$ and thus $\mathcal{A}$, as we are only interested in certain well behaving accumulator values (in particular homomorphisms) and not arbitrary functions. Let us have another look at the definition of $\mathcal{A}$:

```
def 𝒜_acc : (Φ → Φ) → Gdt → Ant Φ
| acc (Gdt.rhs rhs) := Ant.rhs (acc Φ.true) rhs
| acc (Gdt.branch tr1 tr2) := Ant.branch
        (𝒜_acc acc tr1)
        (𝒜_acc ((𝒰_acc acc tr1).and ∘ acc) tr2)
| acc (Gdt.grd (Grd.bang var) tr) := Ant.diverge
        (acc (Φ.var_is_bottom var))
        (𝒜_acc (acc ∘ ((Φ.var_is_not_bottom var).and)) tr)
| acc (Gdt.grd (Grd.tgrd grd) tr) :=
    (𝒜_acc (acc ∘ (Φ.tgrd_in grd)) tr)

def 𝒜 : Gdt → Ant Φ := 𝒜_acc id
```

Since $\mathcal{A}$ is central to many propositions, we define a much simpler function A that does not need an accumulator:

```
def A : Gdt → Ant Φ
| (Gdt.rhs rhs) := Ant.rhs Φ.true rhs
| (Gdt.branch tr1 tr2) := Ant.branch (A tr1) $ (A tr2).map ((U tr1).and)
| (Gdt.grd (Grd.bang var) tr) := Ant.diverge (Φ.var_is_bottom var)
                    $ (A tr).map ((Φ.var_is_not_bottom var).and)
| (Gdt.grd (Grd.tgrd grd) tr) := (A tr).map (Φ.tgrd_in grd)
```

However, $\mathcal{A}$(gdt) is not syntactically equal to A(gdt) for every gdt, as the following counter-example demonstrates:

```
def my_gdt := Gdt.grd (Grd.tgrd (default TGrd))
    (
                (Gdt.rhs (default Rhs))
        .branch (Gdt.rhs (default Rhs))
    )

theorem A_neq_𝒜: (A my_gdt ≠ 𝒜 my_gdt) :=
by simp [A, 𝒜, 𝒜_acc, my_gdt, Ant.map]
```

Instead, we define a semantic on `Ant` Φ and show that A and $\mathcal{A}$ have the same semantic:

```
def Ant.eval_rhss (ant: Ant Φ) (env: Env): Ant bool :=
    ant.map (λ ty, ty.eval env)

theorem A_sem_eq_𝒜 (gdt: Gdt):
    (A gdt).eval_rhss = (𝒜 gdt).eval_rhss
```

When only relying on semantic equivalence, care has to be taken when getting insights on $\mathcal{A}$ by studying A, as `can_prove_empty` does not have to be *well defined* on refinement types modulo semantic equivalence. If two refinement types are semantically equal, `can_prove_empty` could be true for the former, but false for the latter type. A function `can_prove_empty` that is correct and has this well defined property is uncomputable if it returns `true` for the refinement type $\times$ - it would need to return `true` for all refinement types that are empty! Thus, `can_prove_empty` must operate on the refinements of $\mathcal{A}$.

4.2 Redundant RHSs Can Be Removed Without Changing Semantics

Given a guard tree *gdt* with disjoint RHSs and an annotated guard tree *Agdt* that semantically equals `A gdt`, all redundant leaves reported by $\mathcal{R}$ (on *Agdt*, using a correct function `can_prove_empty`) can be removed from *gdt* without changing its semantic. We will later instantiate `Agdt` with $\mathcal{A}$ `gdt`. The indirection introduced

by `Agdt` allows to use the simpler definition of A while `can_prove_empty` still computes emptiness for refinement types in `Agdt` (see chapter 4.1 for why this is important). This internal statement forms the second part of the correctness property defined in chapter 3.2.2 and is formalized as follows:

```
theorem R_red_removable
    (can_prove_empty: CorrectCanProveEmpty)
    { gdt: Gdt } (gdt_disjoint: gdt.disjoint_rhss)
    { Agdt: Ant Φ }
    (ant_def: Agdt.mark_inactive_rhss = (A gdt).mark_inactive_rhss):
        Gdt.eval_option (gdt.remove_rhss (
            (R $ Agdt.map can_prove_empty.val).red.to_finset
        ))
        = gdt.eval
```

The general idea is to focus on a particular but arbitrary environment env: Reasoning about which RHSs can be removed while preserving semantics is much simpler when only considering a single environment.

In fact, we can just evaluate the given guard tree on env and safely remove all RHSs except the one the evaluation ended with. We call RHSs that play no role in the evaluation on env *inactive*, the resulting RHS is called *active*. If the evaluation diverged however, the diverging bang guard must not be removed; thus, all RHSs behind the diverging bang operator except one can be removed. In this case, the bang guard is *active* and all RHSs are inactive. Clearly, at most one node (RHS or bang guard) is active.

The function `Gdt.mark_inactive` directly computes a boolean annotated tree that marks inactive nodes for a given guard tree and environment. The definition of `Gdt.mark_inactive` is very similar to the definition of the denotational semantic of guard trees - this helps proofs that bring these concepts together. This function equals the negation of the semantic of trees annotated with refinement types!

It remains to relate the set of RHSs $r := \mathcal{R}(\mathcal{A}(\text{gdt})).\text{red}$ to the RHSs that can be removed when focusing on a particular environment.

Figure 4.1 sketches the proof idea. Thin arrows mark the data flow, fat arrows the flow of reasoning.

Step 1: Defining gdt and $\mathcal{A}$(gdt)

We start with a guard tree gdt and its annotated tree $\mathcal{A}(\text{gdt})$.

As a detail in the formal proof, we actually use `Agdt` instead of $\mathcal{A}(\text{gdt})$, but since $\text{ant}_2 := \mathcal{A}(\text{gdt}).\text{map}(\neg \circ \Phi.\text{eval}_{\text{env}})$ only depends on the semantic of $\mathcal{A}(\text{gdt})$ and Agdt has the same semantic, this does not change the proof idea.

Step 2: Decomposing $\mathcal{R}$ into R and Ant.map(can_prove_empty), Defining ant_1

To better understand $\mathcal{R}$, we decompose $\mathcal{R}$, which takes an Ant Φ and needs a function `can_prove_empty`, into a function R that takes an Ant bool and a

Figure 4.1: Proof Overview: Redundant RHSs can be removed without changing semantics.



function $f := \text{map}(\text{can\_prove\_empty})$ that computes an Ant bool from an Ant Φ so that $\mathcal{R} = R \circ f$.

In figure 4.1, $\text{ant}_1 := f(\text{gdt})$ represents the object that R works on. Clearly, $\mathcal{R}(\mathcal{A}(\text{gdt})).\text{red} = R(\text{ant}_1).\text{red}$. In the example, only the refinement type associated with leaf 1 is recognized as empty. It is also marked as redundant by $\mathcal{R}/R$.

Step 3: Defining ant_3 and ant_2

ant_3 in figure 4.1 is a boolean annotated tree whose nodes indicate inactivity under env (true if they are inactive, otherwise false). It is much easier to reason about the effect of removing selected RHSs from this tree due to the closely related definitions of `Gdt.mark_inactive` and `Gdt.eval` and the focus on a particular environment, especially if this selection is done by only looking at ant_3.

Is easy to relate ant_1 with ant_3 if we define $\text{ant}_2 := \mathcal{A}(\text{gdt}).\text{map}(\neg \circ \Phi.\text{eval}_{\text{env}})$ as the negation of the evaluation of each refinement type under env.

Step 4: Relating ant_1, ant_2 and ant_3

We can show that each boolean annotation in ant_1 implies ("$\Rightarrow$") the corresponding boolean annotation in ant_2 pointwise (P1): If a refinement type is empty, it must not match.

We can also show $\text{ant}_2 = \text{ant}_3$ (P3), since a node is active if and only if the corresponding refinement type matches.

Step 5: Exploiting the Relationship

It is easy to show that the set of RHSs $R(\text{ant}_3).\text{red}$ can be removed from *gdt* without changing its semantic on env. We hoped that $\mathcal{R}(\text{ant}_\text{a}).\text{red}$ would be a subset of $\mathcal{R}(\text{ant}_\text{b}).\text{red}$ if $\text{ant}_\text{a} \Rightarrow \text{ant}_\text{b}$ to complete the proof. However, this is not the case! See chapter 4.2.1 for a counterexample.

To repair the proof idea, we defined a predicate `is_redundant_set` on sets of RHSs for a given boolean annotated tree. This predicate has the property that if r is a redundant set in ant_a and if $\text{ant}_\text{a} \Rightarrow \text{ant}_\text{b}$, then r is also a redundant set in ant_b (P3).

We show that $R(\text{ant}_1).\text{red}$ is a redundant set (P2) and that redundant sets can be removed from guard trees without changing its semantic (P5). This finishes the proof.

4.2.1 `is_redundant_set`

Given two boolean annotated trees ant_a and ant_b with $\text{ant}_\text{a} \Rightarrow \text{ant}_\text{b}$, we would like to transfer insights on ant_a to ant_b as stated in the previous chapter.

To reason about the removal of redundant RHSs in context of a particular environment, we defined the predicate `is_redundant_set`.

do not read further. Work in progress!

This predicate has the property that if r is a redundant set in ant_a and if $\text{ant}_\text{a} \Rightarrow \text{ant}_\text{b}$, then r is also a redundant set in ant_b (P3).

Later, it will turn out that if the boolean values of that tree indicate *inactivity*, such a redundant set of RHSs can be removed from the guard tree without changing semantics (P5). A node is inactive for a given environment if evaluation does not stop on it. The predicate is defined as following:

```
def Ant.critical_rhs_sets : Ant bool → finset (finset Rhs)
| (Ant.rhs inactive n) := ∅
| (Ant.diverge inactive tr) := tr.critical_rhs_sets ∪ if inactive
    then ∅
    else { tr.rhss }
| (Ant.branch tr1 tr2) := tr1.critical_rhs_sets ∪ tr2.critical_rhs_sets

def Ant.inactive_rhss : Ant bool → finset Rhs
| (Ant.rhs inactive n) := if inactive then { n } else ∅
| (Ant.diverge inactive tr) := tr.inactive_rhss
| (Ant.branch tr1 tr2) := tr1.inactive_rhss ∪ tr2.inactive_rhss

def Ant.is_redundant_set (a: Ant bool) (rhss: finset Rhs) :=
    rhss ∩ a.rhss ⊆ a.inactive_rhss
    ∧ ∀ c ∈ a.critical_rhs_sets, ∃ l ∈ c, l ∉ rhss
```

A redundant set consists of RHSs that are annotated with `false` and avoid critical sets. If a diverge node is annotated with `true`, all its RHSs form a critical set. Each critical set must have one RHS that is not contained in a given redundant set. This ensures that active diverge nodes do not disappear when a redundant set is removed from a guard tree.

We show that all RHSs marked as redundant by $\mathcal{R}$ indeed form a redundant set (P2). We believe that $\mathcal{R}$ actually computes a largest redundant set given a boolean annotated tree. However, a largest redundant set is not unique! In fact, R arbitrarily marks the first redundant RHS as inaccessible to avoid critical sets. It could equally do the same with the last redundant RHS.

We designed redundant sets in a way that

4.3 Accessible RHSs Must Be Detected as Accessible

5 Conclusion

Bibliography

[1] S. Graf, S. Peyton Jones, and R. G. Scott, "Lower your guards: A compositional pattern-match coverage checker," *Proc. ACM Program. Lang.*, vol. 4, Aug. 2020.

[2] "The lean theorem prover (community fork)." `https://github.com/leanprover-community/lean`. Retrieved: 20 Jan. 2021.

[3] L. de Moura, S. Kong, J. Avigad, F. van Doorn, and J. von Raumer, "The lean theorem prover (system description)," in *Automated Deduction - CADE-25* (A. P. Felty and A. Middeldorp, eds.), (Cham), pp. 378–388, Springer International Publishing, 2015.

[4] J. Avigad, L. de Moura, and S. Kong, "Theorem proving in lean." `https://leanprover.github.io/theorem_proving_in_lean/theorem_proving_in_lean.pdf`. Retrieved: 20 Jan. 2021.

[5] "A mathlib overview." `https://leanprover-community.github.io/mathlib-overview.html`. Retrieved: 20 Jan. 2021.

[6] H. Dieterichs, "Lean proof." `https://github.com/hediet/masters-thesis/tree/14833277a507a4f36774d47e965ef1c72811ec41/code`. Retrieved: 20 Jan. 2021.

Erklärung

Hiermit erkläre ich, Henning Dieterichs, dass ich die vorliegende Bachelorarbeit selbstständig verfasst habe und keine anderen als die angegebenen Quellen und Hilfsmittel benutzt habe, die wörtlich oder inhaltlich übernommenen Stellen als solche kenntlich gemacht und die Satzung des KIT zur Sicherung guter wissenschaftlicher Praxis beachtet habe.

Ort, Datum	Unterschrift

Danke

Ich danke meinen Betreuern Sebastian Graf und Sebastian Ullrich, die mich in jeglicher Hinsicht unterstützt haben.